

Optical Layout of Squeezing Experiment

E060180-00-R

Daniel Sigg

June 2006

Initialization

Load Packages / Initialization

```
In[7156]:=
  AppendTo[$Path, "C:\Users\Daniel\ModalModel"];
```

■ General

```
In[7157]:=
  Needs["Utilities`FilterOptions`"];
  Needs["Graphics`Graphics`"];
  Needs["ModalModel`Utilities`"];
```

```
In[7160]:=
  GouyPhase = GuoyPhase;
```

```
In[7161]:=
  $TextStyle = {FontFamily -> "Helvetica", FontSize -> 13};
```

```
In[7162]:=
  plotopt = PlotStyle → {{Thickness [0.007], RGBColor [1, 0, 0]},
    {Thickness [0.007], RGBColor [0, 0, 1]},
    {Thickness [0.007], RGBColor [0.1, 0.7, 0.2]},
    {Thickness [0.007], RGBColor [0.5, 0.5, 0.2]}};
```

Parameters

The parameters are given in SI units.

```
In[7163]:=
  SpeedOfLight = 2.99792458*^8;
  lambda = 1064*^-9;
  kk = N[ $\frac{2\pi}{\text{lambda}}$ ];
  nFusedSilica = 1.44968; (* refractive index *)
```

ABCD Matrices

Definitions for ABCD matrices, see Siegman.

```
In[7167]:=
  Clear[ABCD]
  ABCDPropagator[L_] := {{1, L}, {0, 1}}
  ABCDLens[f_] := {{1, 0}, {- $\frac{1}{f}$ , 1}}
  ABCDPropLens[f_, L_] := {{1, L}, {- $\frac{1}{f}$ , 1 -  $\frac{L}{f}$ }}
  ABCD[A_, B_, C_, D_] := {{A, B}, {C, D}};
  ABCDq[ABCD_?MatrixQ, q_] :=  $\frac{\text{ABCD}[[1, 1]] q + \text{ABCD}[[1, 2]]}{\text{ABCD}[[2, 1]] q + \text{ABCD}[[2, 2]]}$ 
  ABCDz[ABCD_?MatrixQ, q_] := Re[ABCDq[ABCD, q]]
  ABCDz0[ABCD_?MatrixQ, q_] := Im[ABCDq[ABCD, q]]
  ABCDGuoyPhase[ABCD_?MatrixQ, q_] := -Arg[ $\text{ABCD}[[1, 1]] + \frac{\text{ABCD}[[1, 2]]}{q}$ ]
  ABCDSpotSizeRatio[ABCD_?MatrixQ, q_] := Abs[ $\text{ABCD}[[1, 1]] + \frac{\text{ABCD}[[1, 2]]}{q}$ ]
  ABCDGouyPhaseTelescope[rw_, gouy_, z_, z0_] :=
  {{rw (Cos[gouy] -  $\frac{z \text{Sin}[gouy]}{z0}$ )}, rw Sin[gouy] z0 (1 + ( $\frac{z}{z0}$ )2)},
  {- $\frac{\text{Sign}[\text{Sin}[gouy]]}{rw z0}$ , If[Cos[gouy] === 0,  $\frac{\text{Sign}[\text{Sin}[gouy]]}{rw}$ ,  $\frac{1 - \text{Abs}[\text{Sin}[gouy]]}{rw \text{Cos}[gouy]}$ ]}}
```

GouyPhaseTelescope

GouyPhaseTelescope [{zFirst, zLast}, {w0, zw0}, train, k] returns the guoy phase shift acquired in the telescope and the spot size at the output with position zLast, starting with a beam at position zFirst which has a waist size w0 at the waist position zw0. k is the wavevector. Train is a list of doublets consisting of focal length and lens position.

In[7178]:=

```
Clear[GouyPhaseTelescope];
GouyPhaseTelescope[z : {zFirst_, zLast_}, inp : {w0_, zw0_}, train : {{_, _} ..}, k_] :=
Block[{ABCD, qin, i}, ABCD = ABCDPropLens[train[[1, 1]], train[[1, 2]] - zFirst];
For[i = 2, i ≤ Length[train], i++,
ABCD = ABCDPropLens[train[[i, 1]], train[[i, 2]] - train[[i - 1, 2]].ABCD];
ABCD = ABCDPropagator[zLast - train[[-1, 2]].ABCD; qin =  $\frac{1}{2} I k w_0^2 + zFirst - zw_0$ ;
{ABCDGouyPhase[ABCD, qin], ABCDSpotSizeRatio[ABCD, qin] SpotSize[zFirst - zw0, w0, k]}]
GouyPhaseTelescope[z : {zFirst_, zLast_}, inp : {w0_, zw0_}, train : {}, k_] :=
{GouyPhase[zLast - zw0, w0, k] - GouyPhase[zFirst - zw0, w0, k],
SpotSize[zLast - zw0, w0, k]}
```

GPTPlot

GPTPlot [{zFirst, zLast}, {w0, zw0}, train, k] plots the beam profile (spot size) as a function of the propagation direction, starting at the port position and up to the detector position.

In[7181]:=

```
GPTPlot[z : {zFirst_, zLast_}, inp : {w0_, zw0_}, train : {{_, _} ..} | {}, k_, opts___] :=
Block[{opt, int, f, g, plt1, plt2}, opt = FoldList[Lensing[#1, #2, k] &, inp, train];
int = Partition[Join[{zFirst}, (#1[[2]] &) /@ train, {zLast + 1}], 2, 1];
f[x_] = Plus@@Apply[If[#3 ≤ x < #4, 1000 SpotSize[x - #2, #1, k], 0] &,
Flatten/@Transpose[{opt, int}], 2];
g[x_] = Plus@@Apply[Which[#3 ≤ x < #4, GouyPhase[x - #2, #1, k] - GouyPhase[#3 - #2, #1, k],
#3 ≤ x && #4 ≤ x, GouyPhase[#4 - #2, #1, k] - GouyPhase[#3 - #2, #1, k], True, 0] &,
Flatten/@Transpose[{opt, int}], 2]; plt1 = Plot[{f[x], -f[x]},
{x, zFirst, zLast}, DisplayFunction → Identity, opts, MaxBend → 5];
plt2 = Plot[g[x], {x, zFirst, zLast}, DisplayFunction → Identity,
PlotStyle → {RGBColor[0.481, 0.301, 0.023]}, opts];
Show[Graphics[{RGBColor[0.977, 0.121, 0.023], Thickness[0.0075],
(Line[{{#1, -f[#1]}, {#1, f[#1]}]} &) /@ (#1[[2]] &) /@ train}],
Graphics[{RGBColor[0, 0.133, 0.859], Thickness[0.0075],
(Line[{{#1, -f[#1]}, {#1, f[#1]}]} &) /@ z}], plt1, plt2, opts,
Frame → True, GridLines → Automatic, PlotRange → All, AspectRatio → 0.4,
FrameLabel → {"z-position [m]", "spot size [mm]", "Beam Profile",
"Gouy phase [rad]"}, DisplayFunction → $DisplayFunction]
```

TelescopePlot

TelescopePlot[{zFirst, zLast}, {w0, zw0}, train, k] plots the beam profile (spot size) as a function of the propagation direction, starting at the port position and up to the detector position.

In[7516]:=

```
TelescopePlot[z : {zFirst_, zLast_},
  inp : {w0_, zw0_}, train : {{_, _} ..} | {}, k_, opts___] :=
Block[{opt, int, f, g, plt1, plt2}, opt = FoldList[Lensing[#1, #2, k] &, inp, train];
int = Partition[Join[{zFirst}, (#1[[2]] &) /@ train, {zLast + 1}], 2, 1];
f[x_] = Plus@@Apply[If[#3 ≤ x < #4, 1000 SpotSize[x - #2, #1, k], 0] &,
  Flatten /@ Transpose[{opt, int}], 2];
g[x_] = Plus@@Apply[Which[#3 ≤ x < #4, GouyPhase[x - #2, #1, k] - GouyPhase[#3 - #2, #1, k],
  #3 ≤ x && #4 ≤ x, GouyPhase[#4 - #2, #1, k] - GouyPhase[#3 - #2, #1, k], True, 0] &,
  Flatten /@ Transpose[{opt, int}], 2]; plt1 = Plot[{f[x], -f[x]}, {x, zFirst, zLast},
  DisplayFunction → Identity, opts, PlotPoints → 1000, MaxBend → 2];
Show[Graphics[{RGBColor[0.977, 0.121, 0.023], Thickness[0.0075],
  (Line[{{#1, -f[#1]}, {#1, f[#1]}]} &) /@ (#1[[2]] &) /@ train}],
Graphics[{RGBColor[0, 0.133, 0.859], Thickness[0.0075],
  (Line[{{#1, -f[#1]}, {#1, f[#1]}]} &) /@ z}], plt1, opts,
Frame → True, GridLines → Automatic, PlotRange → All, AspectRatio → 0.4,
FrameLabel → {"z-position [m]", "spot size [mm]", "Beam Profile", ""},
DisplayFunction → $DisplayFunction]
```

Parameters

z-axis points from ETM to ITM

z=0 is at the beamsplitter

X is the short arm and perpendicular

ITM/BS are fused silica

```

In[7183]:=
ShortenCavity = 0;
CavityLength = 1.0 - ShortenCavity;
ETMCurvature = 0.52;
ITMCurvature = 0.52;
ITMThickness = 0.028;
BSThickness = 0.025;
gETM = 1 -  $\frac{\text{CavityLength}}{\text{ETMCurvature}}$ ;
gITM = 1 -  $\frac{\text{CavityLength}}{\text{ITMCurvature}}$ ;
gCavity = gETM gITM;
{CavityWaist, CavityRayleighLength, {ETMPosition, ITMPosition}, CavityGouyPhase} =
  GaussianBeam[{CavityLength, -ETMCurvature, ITMCurvature}, kk];
CavityWaistPosition = zzz /. FindRoot[RadiusOfCurvature[zzz, CavityWaist, kk] ==
  ETMCurvature, {zzz, CavityLength/2}]; (* from ETM *)
ETMSpotSize = SpotSize[CavityWaistPosition, CavityWaist, kk];
ITMSpotSize = SpotSize[CavityLength - CavityWaistPosition, CavityWaist, kk];
ITMFocalLength =  $\frac{-\text{ITMCurvature}}{\text{nFusedSilica} - 1}$ ;
MichelsonXLength = 0.3 + ShortenCavity;
MichelsonYLength = 0.4 + ShortenCavity;
MichelsonLength =  $\frac{\text{MichelsonYLength} + \text{MichelsonXLength}}{2}$ ;
MichelsonAsymmetry = MichelsonYLength - MichelsonXLength;
ITMAddedPathLength = (nFusedSilica - 1) ITMThickness;
BSAddedPathLength = (nFusedSilica - 1)  $\sqrt{2}$  BSThickness;
zBS = 0;
zETM = -MichelsonLength - CavityLength;
zITM = -MichelsonLength;
zCavityWaist = zETM + CavityWaistPosition;
{Waist, zWaist} =
  Lensing[{CavityWaist, zCavityWaist}, {ITMFocalLength, zITM}, kk];
RayleighRange = RayleighLength[Waist, kk];

```

In[7208]:=

```

Print["Length of Cavity = ", CavityLength, " m"]
Print["Curvature of ETM = ", ETMCurvature, " m"]
Print["Curvature of ITM = ", ITMCurvature, " m"]
Print["g-Factor of ETM = ", gETM, " m"]
Print["g-Factor of ITM = ", gITM, " m"]
Print["g-Factor of Cavity = ", gCavity, " m"]
Print["Spot Size on ETM = ", 1000 ETMSpotSize, " mm"]
Print["Spot Size on ITM = ", 1000 ITMSpotSize, " mm"]
Print["Rayleigh Length of Cavity Mode = ", CavityRayleighLength, " m"];
Print["Waist Size of Cavity Mode = ", 1000 CavityWaist, " mm"];
Print["Waist Position inside Cavity = ", CavityWaistPosition, " m from ETM"];
Print["Focal Length of ITM = ", ITMFocalLength, " m"];
Print["Michelson Asymmetry = ", MichelsonAsymmetry, " m"];
Print["ETMX Position (for optical path length) = ",
  -CavityLength-MichelsonXLength, " m"];
Print["ITMX Position (for optical path length) = ", -MichelsonXLength, " m"];
Print["ETMY Position (for optical path length) = ",
  -CavityLength-MichelsonYLength, " m"];
Print["ITMY Position (for optical path length) = ", -MichelsonYLength, " m"];
Print["Average ETM Position = ", zETM, " m"];
Print["Average ITM Position = ", zITM, " m"];
Print["Additional ITM optical path length = ", ITMAddedPathLength, " m"];
Print["Additional BS optical path length = ", BSAddedPathLength, " m"];
Print["True ETMX Position = ",
  -CavityLength-MichelsonXLength+ITMAddedPathLength, " m"];
Print["True ITMX Position = ", -MichelsonXLength+ITMAddedPathLength, " m"];
Print["True ETMY Position = ",
  -CavityLength-MichelsonYLength+ITMAddedPathLength+BSAddedPathLength, " m"];
Print["True ITMY Position = ",
  -MichelsonYLength+ITMAddedPathLength+BSAddedPathLength, " m"];
Print["BS Position = ", zBS, " m"];
Print["Waist Size outside Cavity = ", 1000 Waist, " mm"];
Print["Waist Position outside Cavity = ", zWaist, " m"];
Print["Rayleigh Length outside Cavity = ", RayleighRange, " m"];

Length of Cavity = 1. m

Curvature of ETM = 0.52 m

Curvature of ITM = 0.52 m

g-Factor of ETM = -0.923077 m

g-Factor of ITM = -0.923077 m

g-Factor of Cavity = 0.852071 m

Spot Size on ETM = 0.938388 mm

Spot Size on ITM = 0.938388 mm

Rayleigh Length of Cavity Mode = 0.1 m

```

Waist Size of Cavity Mode = 0.184033 mm
Waist Position inside Cavity = 0.5 m from ETM
Focal Length of ITM = -1.15638 m
Michelson Asymmetry = 0.1 m
ETMX Position (for optical path length) = -1.3 m
ITMX Position (for optical path length) = -0.3 m
ETMY Position (for optical path length) = -1.4 m
ITMY Position (for optical path length) = -0.4 m
Average ETM Position = -1.35 m
Average ITM Position = -0.35 m
Additional ITM optical path length = 0.012591 m
Additional BS optical path length = 0.0158986 m
True ETMX Position = -1.28741 m
True ITMX Position = -0.287409 m
True ETMY Position = -1.37151 m
True ITMY Position = -0.37151 m
BS Position = 0 m
Waist Size outside Cavity = 0.128247 mm
Waist Position outside Cavity = -0.702 m
Rayleigh Length outside Cavity = 0.0485625 m

Michelson Compensation

LX is a lens in the X-arm of the Michelson path

LY is a lens in the Y-arm of the Michelson path

```
In[7237]:=
ShiftLY = -0.002;
zLX = -0.125;
fLX = 0.681;
{WaistLX, zWaistLX} =
  Lensing[{Waist, zWaist + MichelsonAsymmetry/2}, {fLX, zLX}, kk];
RayleighRangeLX = RayleighLength[WaistLX, kk];
zLY = zLX - MichelsonAsymmetry + ShiftLY;
fLY = 0.681;
{WaistLY, zWaistLY} =
  Lensing[{Waist, zWaist - MichelsonAsymmetry/2}, {fLY, zLY}, kk];
RayleighRangeLY = RayleighLength[WaistLY, kk];
zLXY =  $\frac{zLX + zLY}{2}$ ;
fLXY =  $\frac{fLX + fLY}{2}$ ;
{WaistLXY, zWaistLXY} =
  Lensing[{Waist, zWaist}, {fLXY, zLXY}, kk];
RayleighRangeLXY = RayleighLength[WaistLXY, kk];
```



```
In[7250]:=
Print["Focal Length of LX Lens = ", fLX, " m"];
Print["Position of LX Lens (for optical path length) = ", zLX, " m"];
Print["True Position of LX Lens = ", zLX, " m"];
Print["Waist Size after LX = ", 1000WaistLX, " mm"];
Print["Waist Position after LX = ", zWaistLX, " m"];
Print["Rayleigh Length after LX = ", RayleighRangeLX, " m"];
Print["Focal Length of LY Lens = ", fLY, " m"];
Print["Position of LY Lens (for optical path length) = ", zLY, " m"];
Print["True Position of LY Lens = ", zLY+BSAddedPathLength, " m"];
Print["Waist Size after LY = ", 1000WaistLY, " mm"];
Print["Waist Position after LY = ", zWaistLY, " m"];
Print["Rayleigh Length after LY = ", RayleighRangeLY, " m"];
Print["Average Waist Size after LX/LY = ", 1000WaistLXY, " mm"];
Print["Average Waist Position after LX/LY = ", zWaistLXY, " m"];
Print["Average Rayleigh Length after LX/LY = ", RayleighRangeLXY, " m"];

Focal Length of LX Lens = 0.681 m

Position of LX Lens (for optical path length) = -0.125 m

True Position of LX Lens = -0.125 m

Waist Size after LX = 0.540863 mm

Waist Position after LX = -2.18306 m

Rayleigh Length after LX = 0.863739 m

Focal Length of LY Lens = 0.681 m

Position of LY Lens (for optical path length) = -0.227 m

True Position of LY Lens = -0.211101 m

Waist Size after LY = 0.534545 mm

Waist Position after LY = -2.25619 m

Rayleigh Length after LY = 0.843678 m

Average Waist Size after LX/LY = 0.537686 mm


Average Waist Position after LX/LY = -2.21956 m

Average Rayleigh Length after LX/LY = 0.853623 m
```

■ Michelson X-arm

In[7265] :=


```
TelescopePlot[{zETM + MichelsonAsymmetry / 2, 5.0},
  {CavityWaist, zCavityWaist + MichelsonAsymmetry / 2},
  {{ITMFocalLength, zITM + MichelsonAsymmetry / 2}, {fLX, zLX}}, kk];
```



■ Michelson Y-arm

In[7266] :=


```
TelescopePlot[{zETM - MichelsonAsymmetry / 2, 5.0},
  {CavityWaist, zCavityWaist - MichelsonAsymmetry / 2},
  {{ITMFocalLength, zITM - MichelsonAsymmetry / 2}, {fLY, zLY}}, kk];
```



■ Michelson X-arm with Lens Position of Y-Arm

In[7267]:=


```
TelescopePlot[{zETM + MichelsonAsymmetry / 2, 5.0},
  {CavityWaist, zCavityWaist + MichelsonAsymmetry / 2},
  {{ITMFocalLength, zITM + MichelsonAsymmetry / 2}, {fLY, zLY}}, kk];
```



■ Michelson Y-arm with Lens Position of X-Arm

In[7268]:=

```
TelescopePlot[{zETM - MichelsonAsymmetry / 2, 5.0},
  {CavityWaist, zCavityWaist - MichelsonAsymmetry / 2},
  {{ITMFocalLength, zITM - MichelsonAsymmetry / 2}, {fLX, zLX}}, kk];
```



Input Beam

Lens L1 is at the input before the probe beam is split off.

```
In[7269]:=
  zL1 = 0.25;
  fL1 = 2.0334;
  fL1 = 1.5346;
  fL1 = 1.146;
  {WaistL1, zWaistL1} =
    Lensing[{WaistLXY, zWaistLXY}, {fL1, zL1}, kk];
  {WaistL1X, zWaistL1X} =
    Lensing[{WaistLX, zWaistLX}, {fL1, zL1}, kk];
  {WaistL1Y, zWaistL1Y} =
    Lensing[{WaistLY, zWaistLY}, {fL1, zL1}, kk];

In[7276]:=
  Print["Focal Length of L1 Lens = ", fL1, " m"];
  Print["Position of L1 Lens = ", zL1, " m"];
  Print["Waist Size after L1 = ", 1000 WaistL1, " mm"];
  Print["Waist Position after L1 = ", zWaistL1, " m"];
  Print["Rayleigh Range after L1 = ", RayleighLength[WaistL1, kk], " m"];
  Print["Waist Size after L1 (X arm Michelson) = ", 1000 WaistL1X, " mm"];
  Print["Waist Position after L1 (X arm Michelson) = ", zWaistL1X, " m"];
  Print["Rayleigh Range after L1 (X arm Michelson) = ",
    RayleighLength[WaistL1X, kk], " m"];
  Print["Waist Size after L1 (Y arm Michelson) = ", 1000 WaistL1Y, " mm"];
  Print["Waist Position after L1 (Y arm Michelson) = ", zWaistL1Y, " m"];
  Print["Rayleigh Range after L1 (Y arm Michelson) = ",
    RayleighLength[WaistL1Y, kk], " m"];

Focal Length of L1 Lens = 1.146 m

Position of L1 Lens = 0.25 m

Waist Size after L1 = 0.391241 mm

Waist Position after L1 = 2.09677 m

Rayleigh Range after L1 = 0.451958 m

Waist Size after L1 (X arm Michelson) = 0.399883 mm

Waist Position after L1 (X arm Michelson) = 2.09954 m

Rayleigh Range after L1 (X arm Michelson) = 0.472143 m

Waist Size after L1 (Y arm Michelson) = 0.382724 mm

Waist Position after L1 (Y arm Michelson) = 2.09328 m

Rayleigh Range after L1 (Y arm Michelson) = 0.432494 m
```

■ Change of L1 Position and focal Length

```
In[7287]:=
```


```
FindMaximum[Lensing[{WaistLXY, zWaistLXY}, {f, zL1}, kk][[2]], {f, 2}]
```

```
Out[7287]=
```

```
{4.24908, {f → 2.05448}}
```


```
In[7288]:=
```

```
Plot[Lensing[{WaistLXY, zWaistLXY}, {fL1, zz}, kk][[2]],  
      {zz, -0.3, 0.3}, GridLines → Automatic, Frame → True, PlotPoints → 1000,  
      FrameLabel → {"Position of L1 (m)", "Waist position (m)"}, PlotRange → All];
```





In[7289]:=


```
Plot[Lensing[{WaistLXY, zWaistLXY}, {f, zL1}, kk][[2]],
  {f, 0, 10}, GridLines → Automatic, Frame → True, PlotPoints → 1000
  , FrameLabel → {"Focal Length of L1 (m)", "Waist position (m)"}, PlotRange → All];
```



In[7290]:=

```
LogLinearPlot[Lensing[{WaistLXY, zWaistLXY}, {f, zL1}, kk][[2]], {f, 0.2, 4},
  GridLines → Automatic, Frame → True, PlotPoints → 1000, FrameLabel →
  {"Focal Length of L1 (m)", "Waist position (m)"}, PlotRange → All, plotopt];
LogLinearPlot[1000 Lensing[{WaistLXY, zWaistLXY}, {f, zL1}, kk][[1]],
  {f, 0.2, 4}, GridLines → Automatic, Frame → True, PlotPoints → 1000,
  FrameLabel → {"Focal Length of L1 (m)", "Waist size (mm)"}, PlotRange → All, plotopt];
LogLinearPlot[RayleighLength[Lensing[{WaistLXY, zWaistLXY}, {f, zL1}, kk][[1], kk],
  {f, 0.2, 4}, GridLines → Automatic, Frame → True, PlotPoints → 1000, FrameLabel →
  {"Focal Length of L1 (m)", "Rayleigh Range (m)"}, PlotRange → All, plotopt];
```






■ Input Beam Profile (Average Michelson)

In[7293]:=


```
TelescopePlot[{zETM, 5.0}, {CavityWaist, zCavityWaist},
  {{ITMFocalLength, zITM}, {fLXY, zLXY}, {fL1, zL1}}, kk];
```



■ Input Beam Profile (X-Arm of Michelson)

In[7294] :=


```
TelescopePlot[{zETM + MichelsonAsymmetry / 2, 5.0},
  {CavityWaist, zCavityWaist + MichelsonAsymmetry / 2},
  {{ITMFocalLength, zITM + MichelsonAsymmetry / 2}, {fLX, zLX}, {fL1, zL1}}, kk];
```



■ Input Beam Profile (Y-Arm of Michelson)

In[7295] :=

```
TelescopePlot[{zETM - MichelsonAsymmetry / 2, 5.0},
  {CavityWaist, zCavityWaist - MichelsonAsymmetry / 2},
  {{ITMFocalLength, zITM - MichelsonAsymmetry / 2}, {fLY, zLY}, {fL1, zL1}}, kk];
```



AS Port Beam

The lens L2 is at the antisymmetric port just after the first turning mirror.

```
In[7296]:=
  zL2 = 0.25;
  fL2 = 1.146;
  {WaistL2, zWaistL2} =
    Lensing[{WaistLXY, zWaistLXY}, {fL2, zL2}, kk]
  {WaistL2X, zWaistL2X} =
    Lensing[{WaistLX, zWaistLX}, {fL2, zL2 - MichelsonAsymmetry/2}, kk]
  {WaistL2Y, zWaistL2Y} =
    Lensing[{WaistLY, zWaistLY}, {fL2, zL2 + MichelsonAsymmetry/2}, kk]

Out[7298]=
  {0.000391241, 2.09677}

Out[7299]=
  {0.000410818, 2.0597}

Out[7300]=
  {0.000372779, 2.13183}
```

```
In[7301]:=
Print["Focal Length of L2 Lens = ", fL2, " m"];
Print["Position of L2 Lens (for optical path length) = ", zL2, " m"];
Print["True Position of L2 Lens = ", zL2 - BSAddedPathLength, " m"];
Print["Waist Size after L2 = ", 1000 WaistL2, " mm"];
Print["Waist Position after L2 = ", zWaistL2, " m"];
Print["Rayleigh Range after L2 = ", RayleighLength[WaistL2, kk], " m"];
Print["Waist Size after L2 (X arm Michelson) = ", 1000 WaistL2X, " mm"];
Print["Waist Position after L2 (X arm Michelson) = ", zWaistL2X, " m"];
Print["Rayleigh Range after L2 (X arm Michelson) = ",
      RayleighLength[WaistL2X, kk], " m"];
Print["Waist Size after L2 (Y arm Michelson) = ", 1000 WaistL2Y, " mm"];
Print["Waist Position after L2 (Y arm Michelson) = ", zWaistL2Y, " m"];
Print["Rayleigh Range after L2 (Y arm Michelson) = ",
      RayleighLength[WaistL2Y, kk], " m"];

Focal Length of L2 Lens = 1.146 m

Position of L2 Lens (for optical path length) = 0.25 m

True Position of L2 Lens = 0.234101 m

Waist Size after L2 = 0.391241 mm

Waist Position after L2 = 2.09677 m

Rayleigh Range after L2 = 0.451958 m

Waist Size after L2 (X arm Michelson) = 0.410818 mm

Waist Position after L2 (X arm Michelson) = 2.0597 m

Rayleigh Range after L2 (X arm Michelson) = 0.49832 m

Waist Size after L2 (Y arm Michelson) = 0.372779 mm


Waist Position after L2 (Y arm Michelson) = 2.13183 m

Rayleigh Range after L2 (Y arm Michelson) = 0.41031 m
```

■ AS Port Beam Profile (Average Michelson)

In[7313]:=


```
TelescopePlot[{zETM, 5.0}, {CavityWaist, zCavityWaist},
  {{ITMFocalLength, zITM}, {fLXY, zLXY}, {fL2, zL2}}, kk];
```



■ AS Port Beam Profile (X-Arm of Michelson)

In[7314]:=


```
TelescopePlot[{zETM + MichelsonAsymmetry / 2, 5.0},
  {CavityWaist, zCavityWaist + MichelsonAsymmetry / 2},
  {{ITMFocalLength, zITM + MichelsonAsymmetry / 2}, {fLX, zLX}, {fL2, zL2}}, kk];
```



■ AS Port Beam Profile (Y-Arm of Michelson)

In[7315]:=

```
TelescopePlot[{zETM - MichelsonAsymmetry/2, 5.0},
  {CavityWaist, zCavityWaist - MichelsonAsymmetry/2},
  {{ITMFocalLength, zITM - MichelsonAsymmetry/2}, {fLY, zLY}, {fL2, zL2}}, kk];
```



Probe Beam

The lens L3 is there to "mirror" the beam on the beam sampler, since we are propagating the wrong way.

The lenses L4 and L5 are for mode matching to the AS port beam.

In[7485]:=

```
zL3 = 0.15; (* beam splitter for probe beam *)
fL3 = ff /. Solve[Lensing[{WaistLXY, zWaistLXY}, {ff, zL3}, kk] [[1]] == WaistLXY, ff] [[1]];
(* Beam propagate from other direction, so just "mirror" it *)
{WaistL3, zWaistL3} = Lensing[{WaistLXY, zWaistLXY}, {fL3, zL3}, kk];
zL4 = 1.291;
fL4 = 0.6806;
{WaistL4, zWaistL4} =
  Lensing[{WaistL3, zWaistL3}, {fL4, zL4}, kk];
zL5 = 1.414;
fL5 = -0.614;
{WaistL5, zWaistL5} =
  Lensing[{WaistL4, zWaistL4}, {fL5, zL5}, kk];
```

■ Find Mode Matching Solution

In[7494]:=

```
ProbeBeamWaist[zzL4_, zzL5_] :=  
  Lensing[Lensing[{WaistL3, zWaistL3}, {fL4, zzL4}, kk], {fL5, zzL5}, kk]; FindRoot[  
  {ProbeBeamWaist[zzL4, zzL5][[1]] == WaistL2, ProbeBeamWaist[zzL4, zzL5][[2]] == zWaistL2},  
  {{zzL4, zL4}, {zzL5, zL5}}, MaxIterations -> 400]
```

Out[7494]=

```
{zzL4 -> 1.29046, zzL5 -> 1.41415}
```

```
In[7495]:=
Print["Focal Length of L3 'Lens' = ", fL3, " m"];
Print["Position of L3 Beam Sampler = ", zL3, " m"];
Print["Waist Size after L3 = ", 1000 WaistL3, " mm"];
Print["Waist Position after L3 = ", zWaistL3, " m"];
Print["Rayleigh Range after L3 = ", RayleighLength[WaistL3, kk], " m"];
Print["Focal Length of L4 Lens = ", fL4, " m"];
Print["Position of L4 Lens = ", zL4, " m"];
Print["Waist Size after L4 = ", 1000 WaistL4, " mm"];
Print["Waist Position after L4 = ", zWaistL4, " m"];
Print["Rayleigh Range after L4 = ", RayleighLength[WaistL4, kk], " m"];
Print["Focal Length of L5 Lens = ", fL5, " m"];
Print["Position of L5 Lens = ", zL5, " m"];
Print["Waist Size after L5 = ", 1000 WaistL5, " mm"];
Print["Waist Position after L5= ", zWaistL5, " m"];
Print["Rayleigh Range after L5 = ", RayleighLength[WaistL5, kk], " m"];

Focal Length of L3 'Lens' = 1.33854 m

Position of L3 Beam Sampler = 0.15 m

Waist Size after L3 = 0.537686 mm

Waist Position after L3 = 2.51956 m

Rayleigh Range after L3 = 0.853623 m

Focal Length of L4 Lens = 0.6806 m

Position of L4 Lens = 1.291 m

Waist Size after L4 = 0.174986 mm

Waist Position after L4 = 1.7694 m

Rayleigh Range after L4 = 0.0904094 m

Focal Length of L5 Lens = -0.614 m

Position of L5 Lens = 1.414 m

Waist Size after L5 = 0.392189 mm


Waist Position after L5= 2.09904 m

Rayleigh Range after L5 = 0.45415 m
```

■ Probe Beam Port Beam Profile (No Mode Matching)

In[7519]:=


```
TelescopePlot[{zETM, 4.0}, {CavityWaist, zCavityWaist},
  {{ITMFocalLength, zITM}, {fLXY, zLXY}, {fL3, zL3}}, kk];
```



■ AS Port Port Beam Profile

In[7518]:=

```
TelescopePlot[{zETM, 4.0}, {CavityWaist, zCavityWaist},
  {{ITMFocalLength, zITM}, {fLXY, zLXY}, {fL2, zL2}}, kk];
```



■ Probe Beam Port Beam Profile with Mode Matching

In[7517]:=

```
TelescopePlot[{zETM, 4.0}, {CavityWaist, zCavityWaist},  
  {{ITMFocalLength, zITM}, {fLXY, zLXY}, {fL3, zL3}, {fL4, zL4}, {fL5, zL5}}, kk];
```

